

LET
(PROGRAMS FOR PRACTICE)

```
5 REM A counting program.
10 LET COUNT=0
20 LET COUNT=COUNT+1
30 PRINT COUNT
40 REM Line 50 says go back to line 20.
50 GOTO 20
```




```
5 REM Practice using numeric variables.
10 LET NUM1=5
20 LET NUM2=7
30 LET NUM3=NUM1+NUM2
40 LET NUM4=NUM1*NUM2
50 ? "NUM1 = "; NUM1
60 ? "NUM2 = "; NUM2
70 ? NUM1;"+";NUM2;"=";NUM3
80 ? NUM1;"*";NUM2;"=";NUM4
```

```
5 REM Practice using string variables.
10 DIM NAME$(15)
20 LET NAME$="-----"
30 ? NAME$;
40 ? " is learning about variables."
```


```
5 REM Using numbers in strings.
10 DIM NAME$(15)
20 LET NAME$="-----"
30 ? "HI! MY NAME IS ";NAME$;". "
40 ? "SEE HOW FAST I CAN COUNT TO 10!"
50 DIM NUMS$(25)
60 LET NUMS$="1 2 3 4 5 6 7 8 9 10"
70 ? NUMS$
```

```
5 REM Assignment without LET; Averaging.
10 N1=7
20 N2=8
30 N3=10
40 N4=105
50 N5=1000
60 N6=(N1+N2+N3+N4+N5)/5
70 ? "The average of "; N1;"", ";N2;"", ";N3;"",
    ", ";N4;"", and ";N5;" " is ";N6;"."
```

VARIABLES - ACTIVITY #2



DIM NAME\$(5)



DIM ANSWER\$(9)

```
DIM NAME$(6)
NAME$="GEORGE"
PRINT NAME$
```

```
NAME$="CAROLE"
PRINT NAME$
```

PROGRAMMING CHALLENGES USING VARIABLES

Choose one the the following tasks and write a program that completes it.

1. Ask the user to enter five words. Print the nursery rhyme below using the words as input. Specify the kinds of words you want, so that when you use them to fill in the blanks, the nursery rhyme will make sense.

```
Little Miss _____  
Sat on a _____  
Eating her _____ and _____.  
Along came a _____  
Who sat down beside her  
And frightened Miss _____ away.
```

2. Compute a player's batting average when given the number of times at bat, number of hits and number of walks.

3. Save yourself time by writing one form letter that you can send to several people by just changing the greeting. If you want to get fancy, you could also change some of the words in the letter. The following is a short example:

Dear _____,

How are you, _____? I am having a great time
at Atari Summer Camp.

_____,
Sandy

This could also be used to send a "personalized" message to everyone in camp.

4. Ask the user to enter the appropriate dimensions and then compute the area of a geometric figure such as a square, rectangle, triangle, circle, or parallelogram.

5. Use INPUT, LET, at least 3 string variables, and 2 numeric variables to write a program on a topic of your choosing.

INPUT
CAMPER COPY

```
5 REM Asks for a name. Uses string input.
10 DIM NAME$(20)
20 ? "What is your first name";
30 INPUT NAME$
40 ? "HI, ";NAME$;". I'm happy to make
    your acquaintance."
```

```
5 REM Averaging program with input.
10 ? "This program will calculate the average"
20 ? "of any five numbers you choose. Type the"
30 ? "numbers in with commas between them--"
40 ? "like this 10,3,40,70,90."
50 INPUT NUM1,NUM2,NUM3,NUM4,NUM5
60 AVE=(NUM1+NUM2+NUM3+NUM4+NUM5)/5
70 ? "The average of your numbers is ";AVE;"."
```

```
5 REM Calculates age in the year 2001
10 ? "Have you ever wondered how old you"
20 ? "will be in the year 2001?"
30 ? "This program will do the necessary"
40 ? "calculations for you very quickly."
50 ? "Enter your age:";
60 INPUT AGE
70 ? "What year is it? (Be sure to use all"
80 ? "four numbers, eg. 1983)";
90 INPUT YEAR
100 ANSWER=2001-YEAR+AGE
110 ? "In 2001, you will be ";ANSWER;" years old."
```

INPUT
(CAMPER COPY CONTINUED)

```
5 REM This program allows the user to
10 REM change the hue and luminance values
20 REM in a SETCOLOR command.
30 ? "In this program you will be able to"
40 ? "change the color of the background on"
50 ? "the screen."
60 ? "Choose a number (0-15) for the hue";
70 INPUT HUE
80 ? "Choose a number (0-14) for the luminance";
90 INPUT LUM
100 GR. 3
110 SETCOLOR 4,HUE,LUM
```

```
5 REM This program allows the user to
10 REM change the graphics mode to
20 REM display different sizes of text on
30 REM on the screen. It also allows the
40 REM user to change the text on the screen.
50 ? "In this program, you will be able to"
60 ? "enter a number from 0 to 2 for the"
70 ? "graphics mode and a word that you"
80 ? "want to display."
90 ? "What graphics mode would you like";
100 INPUT MODENUM
120 ? "You may display any word up to 15 letters."
130 ? "What word would you like to display";
140 DIM WORD$(15)
150 INPUT WORD$
160 GR. MODENUM
170 COLOR 2
180 POS. 5,6
190 ? #6;WORD$
```

INPUT
(CAMPER COPY CONTINUED)

```
5 REM This program uses string variables to
10 REM create a sentence generator.
20 DIM NOUN$(15),VERB$(15),
   ADJ$(15),NOUN2$(15)
30 ? "Please enter a plural noun:";:INPUT NOUN$
40 ? "Now enter a verb that goes with a"
50 ? "plural noun:";:INPUT VERB$
60 ? "Next enter an adjective. Remember"
70 ? "that an adjective is a word"
80 ? "that describes a noun:";:INPUT ADJ$
90 ? "Finally, enter another plural noun:";
100 INPUT NOUN2$
110 ? "Did you know that ";
120 ? NOUN$;" ";VERB$;" ";
130 ? ADJ$;" ";NOUN2$;"?"
```

NUMBER PUZZLE

```

10 REM *** Number Pattern Puzzle ***
20 DIM ANS$(1),MORE$(1)
30 ? ">":? :?
40 ? "This is a number puzzle. Try"
50 ? "to figure out what the next"
60 ? "number will be in each sequence"
70 ? "These are the sequences:";?
80 ? "A. 1, 3, 5, 7, 11, 13, 17, ___"
90 ? "B. 40, 51, 62, 73, 84, 95, ___"
100 ? "C. 1, 1, 2, 3, 5, 8, 13, 21, ___"
110 ? "D. 3, 3, 5, 4, 4, 3, 5, 5, ___";?
120 ? "What sequence would you like"
130 ? "to try? (Type a letter.)";
140 INPUT ANS$
150 IF ANS$="A" THEN GOSUB 5000
160 IF ANS$="B" THEN GOSUB 5100
170 IF ANS$="C" THEN GOSUB 5200
180 IF ANS$="D" THEN GOSUB 5300
200 ? :? "Do you want to try another"
210 ? "sequence? (Type Y or N)";
220 INPUT MORE$
230 IF MORE$="Y" THEN ? ">":GOTO 70
240 END
5000 REM ****prime number sequence***
5005 REM ****
5010 ? ">":? :? :REM * clears screen *
5020 ? "Type a number to finish this"
5025 ? "sequence. You get three chances.";?
5030 ? "1, 3, 5, 7, 11, 13, 17, ___"
5040 FOR COUNT=1 TO 3
5050 ? :? "NUMBER";
5060 INPUT NUM
5070 IF NUM=19 THEN ? "That's correct.":RETURN
5080 ? "It's not ";NUM;". These are prime numbers."
5090 NEXT COUNT
5095 RETURN

```

WHO AM I?

```

2300 REM ***      Blaise Pascal      ***
2310 ? ">":PRINT
2320 ? "A French mathematician who was"
2330 ? "the first person to invent a"
2340 ? "significant calculating "
2350 ? "machine."
2360 PRINT :PRINT
2370 RETURN
2400 REM ***      Boole      ***
2410 ? ">":PRINT
2420 ? "An English logician. The"
2430 ? "pioneer of modern symbolic logic."
2440 PRINT :PRINT
2450 RETURN
2500 REM ***      Hollerith      ***
2510 ? ">":?
2520 ? "An American inventor. The"
2560 ? "first to do a practical"
2570 ? "implementation of punched cards."
2580 PRINT :PRINT
2590 RETURN
2600 REM **  Who Am I - Directions **
2610 ? ">":? :?
2620 ? "      WHO AM I?"
2625 ?
2630 ? "After you read the description"
2640 ? "of the person, choose your"
2650 ? "answer from the names given."
2660 ? "Type in the letter of the"
2670 ? "correct answer. Each of the"
2680 ? "names is famous in computing."
2685 ? "Press RETURN when you are "
2690 ? "ready to begin."
2695 INPUT R$
2699 RETURN
2700 REM ***  Asks for answer      ***
2710 REM xinput with correct answer*
2720 ? :? "Who am I? Type a letter";
2730 INPUT USERANS$
2740 IF USERANS$=CORRECTANS$ THEN ? "That's correct.":RETURN
2750 ? "That's not my name. The"
2760 ? "correct answer is ";CORRECTANS$;". "
2770 RETURN
2800 REM ***      Answers      ***
2810 ? :? :?
2820 ? "      A. Pascal"
2830 ? "      B. Hollerith"
2840 ? "      C. Ada"
2850 ? "      D. Boole"
2860 ? "      E. Babbage"
2870 RETURN
29000 REM ** Delay Loop **
29010 FOR WAIT=1 TO 500:NEXT WAIT
29020 RETURN

```


WHO AM I?

```

10 REM *Who Am I-Names in Computing*
20 DIM CORRECTANS$(1),USERANS$(1),R$(1)
30 GOSUB 2600:REM **   Directions   **
40 GOSUB 2100:REM **   Babbage     **
50 GOSUB 2800:REM **   Answers     **
60 CORRECTANS$="E"
70 GOSUB 2700:REM ** Answer Input **
80 GOSUB 29000:REM ** Wait Loop   **
90 GOSUB 2500:REM ** Hollerith   **
100 GOSUB 2800:REM ** Answers     **
110 CORRECTANS$="B"
120 GOSUB 2700:REM ** Answer Input**
130 GOSUB 29000:REM ** Wait Loop   **
140 GOSUB 2200:REM **   Ada       **
150 GOSUB 2800:REM **   Answers     **
160 CORRECTANS$="C"
170 GOSUB 2700:REM ** Answer Input**
180 GOSUB 29000:REM ** Wait Loop   **
190 GOSUB 2300:REM **   Pascal     **
200 GOSUB 2800:REM **   Answers     **
210 CORRECTANS$="A"
220 GOSUB 2700:REM ** Answer Input**
230 GOSUB 29000:REM ** Wait Loop   **
240 GOSUB 2400:REM **   Boole      **
250 GOSUB 2800:REM **   Answers     **
260 CORRECTANS$="D"
270 GOSUB 2700:REM ** Answer Input**
280 GOSUB 29000:REM ** Wait Loop   **
290 END
2100 REM ***   Charles Babbage   ***
2110 ? ">":PRINT
2120 ? "An English mathematician and"
2130 ? "inventor who is often called"
2140 ? "the Father of Computing. He"
2150 ? "said, 'I am thinking that all"
2160 ? "those tables might be"
2170 ? "calculated by machinery.'"
2180 ? :?
2190 RETURN
2200 REM *****   ADA   *****
2210 ? ">":PRINT
2220 ? "An exceptional English mathematician"
2230 ? "who is credited with being the first"
2240 ? "person to make the statement that"
2250 ? "computers can do only what you"
2260 ? "program them to do. Wrote about"
2270 ? "Babbage's Analytical Engine."
2280 PRINT :PRINT
2290 RETURN

```

FUNCTIONS RND & INT
(CAMPER COPY)

```
PRINT RND(0)*4  
PRINT RND(0)*10  
PRINT RND(0)*50
```

```
PRINT INT(1.9)  
PRINT INT(30.111)  
PRINT INT(0.65)  
PRINT INT(100000.9)
```

```
10 NUM=RND(0)*5  
20 PRINT NUM  
30 GOTO 10
```

```
10 NUM=RND(0)*5+1
```

Run the program again and note the difference in the output.
Change line 10 again to:

```
10 NUM=INT(RND(0)*5+1)
```

IF...THEN
CAMPER COPY

```
10 REM Example of a conditional
20 ? "␣"; REM * Clears screen *
30 ?:"Type in a number: ";
40 INPUT NUM
50 IF NUM>10 THEN ? "That's too big."
60 IF NUM<10 THEN ? "That's too small."
70 IF NUM=10 THEN ? "That's the number I had
in mind.":GOTO 90
80 GOTO 30
90 END
```

```
10 REM ** Coin Toss **
20 DIM R$(1)
30 ? "␣";REM Clears screen
40 ? "One person chooses heads, the"
50 ? "other person chooses tails.":?
60 ? "Press return when you have decided.";
70 INPUT R$: ?
80 IF INT(2*(RND(1)))<1 THEN ? "The person
who chose heads goes first.":GOTO 100
90 ? "The person who chose tails goes first."
100 GOTO 100
```

```
10 REM * A matter of taste *
20 REM * Using strings in IF...THEN blocks *
30 DIM ANS$(3)
40 ? "␣";REM * Clears screen *
50 ? "Do you like chocolate? (Type YES or NO)"
60 INPUT ANS$
70 IF ANS$="YES" THEN ? "You have good taste.":
GOTO 90
80 ? "Your taste is questionable!"
90 END
```

FORTUNE TELLER:

```

10 REM Fortune teller
20 PRINT ">":REM Clear screen
30 PRINT :PRINT "I will tell you your fortune."
40 PRINT "Let's see...":PRINT
50 NUM=INT(3*RND(0))
60 FOR WAIT=1 TO 1000:NEXT WAIT
70 PRINT "+ + + + + + + + + + +"
80 PRINT "You will become very ";
90 IF NUM=0 THEN PRINT "rich."
100 IF NUM=1 THEN PRINT "poor."
110 IF NUM=2 THEN PRINT "powerful."
120 NUM=INT(3*RND(0))
130 FOR WAIT=1 TO 750:NEXT WAIT
140 PRINT "You will also be very ";
150 IF NUM=0 THEN PRINT "happy."
160 IF NUM=1 THEN PRINT "famous."
170 IF NUM=2 THEN PRINT "popular."
180 PRINT "+ + + + + + + + + + +"
190 GOTO 30

```

WRITING A PROGRAM (CAMPER COPY)

1. General description of the problem.

Write a short story about a person who leaves the city to go to the beach on a vacation.

2. Be more specific about what the program will do by writing out the story.

A man named Fred was very bored with life. It seemed like all he ever did was go around in circles. One day he decided to leave the big city. He got on a train and went to the beach for a vacation. The day he arrived, it was sunny and warm. The sound of the ocean was very calming to his nerves. However...That night a storm came up and it rained and it rained and it rained. He decided to phone home to see if the weather was any better there. But since his children were always on the phone, all he got was a busy signal. Just as he was at the end of his rope and ready to return home, the birds began to sing, the sun came out, and he....

3. Decide what pictures and sound effects might be appropriate and make a list of them. Then determine which ones are available in the subroutine library and which ones need to be written.

4. Divide the story into sections that will fit with the sound and graphics subroutines.

WRITING A PROGRAM

(CAMPER COPY - CONTINUED)

5. Write out English statements that show the solution step by step.

Title screen
Author screen
Introduce the main character and his problem
Graphics routine for going around in circles
Leaves the city on a train
Train sound effect
Arrives at the beach
Ocean sound effect
A storm comes up and it starts raining
Graphics routine for rain
It rains some more
Graphics routine for rain
It rains some more
Graphics routine for rain
He decides to phone home to ask about the weather
Busy signal sound effect
He is at the end of his rope when birds sing
Birds chirping sound effect
The sun comes out
Graphics routine for sun
And he...To be continued.

VACATION - MAIN PROGRAM

```
10 REM ***** MAIN PROGRAM *****
50 GOSUB 10300:REM * Title Page *
55 GOSUB 29000:REM * Wait Loop *
60 GOSUB ____:REM * Author Page *
65 GOSUB ____:REM * Wait Loop *
70 GOSUB 1100:REM * Introduction *
80 GOSUB ____:REM * Wait Loop *
90 GOSUB ____:REM *Going in Circles*
100 GOSUB 1200:REM *Leaving the city *
110 GOSUB ____:REM * Train Sound *
120 GOSUB 1300:REM * Arrive at beach *
130 GOSUB ____:REM * Ocean Sound *
140 GOSUB 1400:REM * The Storm *
150 GOSUB ____:REM * Wait Loop *
160 GOSUB ____:REM * Rain Graphic *
170 GOSUB 1700:REM * Rain text *
180 GOSUB ____:REM * Wait Loop *
190 GOSUB ____:REM * Rain graphic *
200 GOSUB 1700:REM * Rain text *
210 GOSUB ____:REM * Wait Loop *
--- Graphics routine for rain drops
230 GOSUB 1500:REM * Phone Home *
--- Sound effect for telephone busy signal
250 GOSUB 1800:REM * End of Rope *
--- Sound effect of birds chirping
270 GOSUB 1900:REM * Sunshine text *
--- Delay loop to keep text on screen
--- Graphics routine for sun shining
295 GOSUB 29000:REM * Wait Loop *
300 GOSUB 2000:REM * Continued text *
--- Delay loop to keep text on screen.
350 END
360 REM
370 REM
380 REM
```

LIST OF SUBROUTINES

```
INTRO.TXT
LEAVING.TXT
REACH.TXT
STORM.TXT
RAIN.TXT
PHONHOME.TXT
ENDROPE.TXT
SUNSHINE.TXT
CONTINUE.TXT
```

WRITING A PROGRAM

(CONTINUED)

2. The subroutines listed below are on the BASIC Utility Disk. Enter and store them using the following steps exactly as they are written. They put the text on the screen. The information in parentheses is the REMark in the main program.

Use these steps for each subroutine.

1. Type NEW to clear memory.
2. Put in the BASIC Utility Disk.
3. Type ENTER"D:_____ ", putting the subroutine name in the blank.
4. Put in the camper's personal disk.
5. Type LIST"D:_____ ", putting the subroutine name in the blank.
6. GOTO 1.

LIST OF SUBROUTINES

INTRO.TXT
LEAVING.TXT
BEACH.TXT
STORM.TXT
RAIN.TXT
PHONHOME.TXT
ENDROPE.TXT
SUNSHINE.TXT

Activity #4

Putting it all together.

1. Type in the main program.
2. Print a directory of the camper's disk to get a list of subroutines.
3. ENTER all subroutines. DO NOT TYPE NEW AFTER EACH ENTRY!
4. Run the program and debug it.
5. Save the program using SAVE"D:_____ ", putting a name the camper chooses in the blank.

WRITING A PROGRAM

(CONTINUED)

Activity #5

Complete at least one of the challenges listed below.

1. Finish the story by adding text, sound, and graphics subroutines.
2. Rearrange the main program, so that action happens in a different sequence.
3. Use entries in the subroutine library to write an original program that tells a story.

JOYSTICKS AND CONTROLLERS CAMPER COPY

```
10 NUM=STRIG(0)
20 PRINT NUM
30 GOTO 10
```

```
10 REM * Joystick and SOUND
20 PITCH=STICK(0)
30 SOUND 0,PITCH*10,10,10
40 GOTO 10
```

```
10 REM * Using a paddle *
20 PITCH=PADDLE(0)
30 PRINT PITCH
40 SOUND 0,PITCH,10,10
50 GOTO 10
```

```
10 REM *Experimenting with paddles and sound.*
20 VOL=10
30 SOUND 0,PADDLE(0),10,VOL
40 IF PTRIG(0)=0 THEN GOSUB 100
50 FOR DELAY=1 TO 40:NEXT DELAY
60 GOTO 30
100 VOL=VOL+1
110 IF VOL=16 THEN VOL=0
120 RETURN
```

Add these lines:

```
25 PRINT PADDLE(0)
105 PRINT ,VOL
```

EXPLORE LPRINT

1. Be sure the printer and the interface are on. Type in the program below.

```
10 REM This program illustrates what LPRINT does
20 PRINT "THIS LINE WILL OUTPUT TO THE SCREEN"
30 LPRINT "THIS LINE WILL NOT OUTPUT TO THE SCREEN"
40 LPRINT "LPRINT WORKS JUST LIKE PRINT"
50 LPRINT "EXCEPT THAT IT PRINTS ON THE PRINTER"
```

Run the program. Look carefully at the output to see which lines were output to the screen and which ones were output to the printer.

2. Type in the following program and then run it to further illustrate LPRINT.

```
10 GRAPHICS 2:SETCOLOR 4,13,2
20 LPRINT "There was a young man from Purdue"
30 POSITION 5,1:PRINT #6; "#####"
40 LPRINT "Who dreamed he was eating rock stew."
50 POSITION 5,2:PRINT #6; "#The printer#"
60 LPRINT "    He woke up in the night,"
70 POSITION 5,3:PRINT #6; "#is printing#"
80 LPRINT "    With a terrible fright";
90 POSITION 5,4:PRINT #6; "##a limerick##"
100 LPRINT "To find it was perfectly true."
110 POSITION 5,5:PRINT #6; "#####"
```

3. Combine PRINT and LPRINT in a program. An example is given below.

```
10 PRINT "SCREEN": LPRINT "PRINTER"
20 PRINT "FIRST": LPRINT "SECOND"
30 PRINT "5 4 3 2 1",
40 LPRINT "5 4 3 2 1",
50 PRINT "BLASTOFF!"
60 LPRINT "BLASTOFF!"
```

Be creative and make up your own program. Save the program on your diskette.

LPRINT

```
10 REM This program illustrates what LPRINT does
20 PRINT "THIS LINE WILL APPEAR ON THE SCREEN"
30 LPRINT "THIS LINE WILL NOT APPEAR ON THE SCREEN"
40 LPRINT "LPRINT WORKS JUST LIKE PRINT"
50 LPRINT "EXCEPT THAT IT PRINTS ON THE PRINTER"
```

```
10 LPRINT "Said a young, but wise robot
   named Truman,"
20 PRINT "The instructor's supremacy fades"
30 LPRINT "'When a man starts fussin'
   and fumin',"
40 PRINT "When robots become teaching aides"
50 LPRINT "   And is clumsy and coarse"
60 PRINT "   And students bring treats"
70 LPRINT "   I think of the source,"
80 PRINT "   Of candy and sweets,"
90 LPRINT "And remember he is only human.'"
100 PRINT "To the robot who makes out the
   grades."
```

```
10 PRINT "SCREEN": LPRINT "PRINTER"
20 PRINT "FIRST": LPRINT "SECOND"
30 PRINT "5 4 3 2 1",
40 LPRINT "5 4 3 2 1",
50 PRINT "BLASTOFF!"
60 LPRINT "BLASTOFF!"
```

USING AND

```
10 REM *** Using AND ***
20 DIM AN$(10),CORRECT$(1)
30 CORRECT$="N":COUNT=0
40 ? "3"?
50 ? "Who published 'On Computable"
60 ? "Numbers', one of the most important"
70 ? "papers in the foundations of"
80 ? "computer science? You get three"
90 ? "chances to give the right answer."?:
100 INPUT AN$
110 IF AN$="Turing" THEN CORRECT$="Y"
120 COUNT=COUNT+1
130 ? "COUNT = ";COUNT
140 ? "CORRECT = ";CORRECT$?:
150 IF COUNT<3 AND CORRECT$="N" THEN ? "Try again: ";GOTO 100
160 IF CORRECT$="Y" THEN ? "Yes, it was Alan Matheson Turing."?:GOTO 200
170 ? "This was a difficult question. The"
180 ? "answer can be found on page 79 of"
190 ? "the book, 'The Making of the Micro'."
200 END
```

OR / AND CAMPER COPY

```
10 REM * Use of OR and AND *
20 ? "↵":REM Clear screen
30 DIM COLOR$(15)
40 ? "Type in your favorite color. Use"
50 ? "all capital letters like this, RED."?:
60 INPUT COLOR$
70 IF COLOR$="RED" OR COLOR$="YELLOW" THEN
? "That is one of my favorites, too!":GOTO 90
80 ? "Your taste in colors is different
than mine."
90 END
```


```
30 DIM COLOR1$(15), COLOR2$(15)
40 ? "Type in two colors. Use"
60 ? "What is the first color";
70 INPUT COLOR1$
80 ? "What is the second color";
90 INPUT COLOR2$
100 IF COLOR1$="RED" AND COLOR2$="YELLOW" THEN
? "Those are my favorite colors.":GOTO 120
110 ? "I'm not fond of those colors."
120 END
```

```
10 REM Use of OR to solve one input problem
20 ? "↵":REM Clears screen
30 DIM ANS$(3)
40 ? "Is your favorite car a Porsche";
50 INPUT ANS$
60 IF ANS$="YES" OR ANS$="Y" OR ANS$="yes"
OR ANS$="y" OR ANS$="Yes" THEN ? "What class!":
GOTO 80
70 ? "It isn't. I'm amazed!"
80 END
```



GRAPHICS MODE 6 or 7

ATARI SPECIAL PROJECTS




NOTES



GRAPHICS MODE 4 or 5

ATARI SPECIAL PROJECTS




NOTES



GRAPHICS MODE 3

ATARI SPECIAL PROJECTS



NOTES




8

11




GRAPHICS MODE 1



NOTES

Control Graphics Keyboard



ERROR CODES

ERROR CODE ERROR CODE MESSAGE

2	Memory Insufficient
3	Value Error
4	Too Many Variables
5	String Length Error
6	Out of Data Error
7	Number greater than 32767
8	Input Statement Error
9	Array or String DIM Error
10	Argument Stack Overflow
11	Floating Point Overflow/ Underflow Error
12	Line Not Found
13	No Matching FOR Statement
14	Line Too Long Error
15	GOSUB or FOR Line Deleted
16	RETURN Error
17	Garbage Error
18	Invalid String Character

ERROR CODE ERROR CODE MESSAGE

142	Serial Bus Data Frame Overrun
143	Serial bus data frame checksum error
144	Device done error
145	Read after write compare error
146	Function not implemented
147	Insufficient RAM
160	Drive number error
161	Too many OPEN files
162	Disk full
163	Unrecoverable system data I/O error
164	File number mismatch
165	File name error
166	POINT data length error
167	File locked
168	Command invalid
169	Directory full
170	File not found
171	POINT invalid

Note: The following are INPUT/OUTPUT errors that result during the use of disk drives, printers, or other accessory devices. Further information is provided with the auxiliary hardware.

19	LOAD program Too Long
20	Device Number Larger
21	LOAD File Error
128	BREAK Abort
129	IOCB
130	Nonexistent Device
131	IOCB Write Only
132	Invalid Command
133	Device or File not Open
134	BAD IOCB Number
135	IOCB Read Only Error
136	EOF
137	Truncated Record
138	Device Timeout
139	Device NAK
140	Serial Bus
141	Cursor Out of Range